

G01FEF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G01FEF returns the deviate associated with the given lower tail probability of the beta distribution, via the routine name.

2 Specification

```

real FUNCTION G01FEF(P, A, B, TOL, IFAIL)
  INTEGER                IFAIL
  real                   P, A, B, TOL

```

3 Description

The deviate, β_p , associated with the lower tail probability, p , of the beta distribution with parameters a and b is defined as the solution to:

$$P(B \leq \beta_p : a, b) = p = \frac{\Gamma(a+b)}{\Gamma(a)\Gamma(b)} \int_0^{\beta_p} B^{a-1}(1-B)^{b-1} dB, \quad 0 \leq \beta_p \leq 1; a, b > 0.$$

The algorithm is a modified version of the Newton–Raphson method, following closely that of Cran *et al.* [1].

An initial approximation, β_0 , to β_p is found (see [1]), and the Newton–Raphson iteration

$$\beta_i = \beta_{i-1} - \frac{f(\beta_{i-1})}{f'(\beta_{i-1})}$$

where $f(\beta) = P(B \leq \beta : a, b) - p$ is used, with modifications to ensure that β remains in the range (0,1).

4 References

- [1] Cran G W, Martin K J and Thomas G E (1977) Algorithm AS109. Inverse of the incomplete beta function ratio *Appl. Statist.* **26** 111–114
- [2] Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworths

5 Parameters

- 1: P — *real* *Input*
On entry: the probability, p , from the required beta distribution.
Constraint: $0.0 \leq P \leq 1.0$.
- 2: A — *real* *Input*
On entry: the first parameter, a , of the required beta distribution.
Constraint: $0.0 < A \leq 10^6$.
- 3: B — *real* *Input*
On entry: the second parameter, b , of the required beta distribution.
Constraint: $0.0 < B \leq 10^6$.

4: TOL — *real* *Input*

On entry: the relative accuracy required by the user in the result. If G01FEF is entered with TOL greater than or equal to 1.0 or less than 10 times *machine precision* (see X02AJF), then the value of $10 \times \textit{machine precision}$ is used instead.

5: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL \neq 0 on exit, users are recommended to set IFAIL to -1 before entry. **It is then essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings specified by the routine:

If on exit IFAIL = 1 or 2, then G01FEF returns 0.0.

IFAIL = 1

On entry, $X < 0.0$,
or $X > 1.0$.

IFAIL = 2

On entry, $A \leq 0.0$,
or $A > 10^6$,
or $B \leq 0.0$,
or $B > 10^6$.

IFAIL = 3

There is doubt concerning the accuracy of the computed result. 100 iterations of the Newton-Raphson method have been performed without satisfying the accuracy criterion (see Section 7). The result should be a reasonable approximation of the solution.

IFAIL = 4

Requested accuracy not achieved when calculating beta probability. The result should be a reasonable approximation to the correct solution. The user should try setting TOL larger.

7 Accuracy

The required precision, given by TOL, should be achieved in most circumstances.

8 Further Comments

The typical timing will be several times that of G01EEF and will be very dependent on the input parameter values. See G01EEF for further comments on timings.

9 Example

Lower tail probabilities are read for several beta distributions, and the corresponding deviates calculated and printed, until the end of data is reached.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G01FEF Example Program Text
*      Mark 14 Release.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real            A, B, P, TOL, X
      INTEGER          IFAIL
*      .. External Functions ..
      real            G01FEF
      EXTERNAL         G01FEF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G01FEF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Probability      A          B      Deviate'
      WRITE (NOUT,*)
20     READ (NIN,*,END=40) P, A, B
      TOL = 0.0e0
      IFAIL = -1
*
      X = G01FEF(P,A,B,TOL,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
          WRITE (NOUT,99999) P, A, B, X
      ELSE
          WRITE (NOUT,99999) P, A, B, X, ' NOTE: IFAIL = ', IFAIL
      END IF
      GO TO 20
40     STOP
*
99999 FORMAT (1X,F9.4,2F10.3,F10.4,A,I1)
      END

```

9.2 Program Data

```

G01FEF Example Program Data.
0.5000  1.0  2.0          :P A B
0.9900  1.5  1.5          :P A B
0.2500 20.0 10.0          :P A B

```

9.3 Program Results

G01FEF Example Program Results

Probability	A	B	Deviate
0.5000	1.000	2.000	0.2929
0.9900	1.500	1.500	0.9672
0.2500	20.000	10.000	0.6105